
Modules

A Haskell module is a collection of related functions, types and
typeclasses.

A Haskell program is a collection of modules where the main
module loads up the other modules and then uses the functions
defined in them to do something.

Having code split up into several modules has quite a lot of
advantages. If a module is generic enough, the functions it exports
can be used in a multitude of different programs. If your own code
is separated into self-contained modules which don’t rely on each
other too much (we also say they are loosely coupled), you can
reuse them later on.

It makes the whole deal of writing code more manageable by having
it split into several parts, each of which has some sort of purpose.



Modules

The Haskell standard library is split into modules, each of them
contains functions and types that are somehow related and serve
some common purpose.

There’s a module for manipulating lists, a module for concurrent
programming, a module for dealing with complex numbers, etc.

All the functions, types and typeclasses that we’ve dealt with so far
were part of the Prelude module, which is imported by default.

In this chapter, we’re going to examine a few useful modules and
the functions that they have.



Modules

The syntax for importing modules in a Haskell script is
import <module name>.

This must be done before defining any functions, so imports are
usually done at the top of the file.

One script can, of course, import several modules. Just put each
import statement into a separate line.



Modules

import Data.List

numUniques :: (Eq a) => [a] -> Int

numUniques = length . nub

When you do import Data.List, all the functions that
Data.List exports become available in the global namespace,
meaning that you can call them from wherever in the script.

nub is a function defined in Data.List that takes a list and weeds
out duplicate elements.

Composing length and nub by doing length . nub produces a
function that’s the equivalent of \xs -> length (nub xs).



Importing modules

You can also put the functions of modules into the global
namespace when using GHCI. If you’re in GHCI and you want to
be able to call the functions exported by Data.List, do this:

Prelude> :m + Data.List

Prelude Data.List>

If we want to load up the names from several modules inside
GHCI, we don’t have to do :m + several times, we can just load up
several modules at once.

Prelude> :m + Data.List Data.Map Data.Set

Prelude Data.List Data.Map Data.Set>

However, if you’ve loaded a script that already imports a module,
you don’t need to use :m + to get access to it.



Importing modules

If you just need a couple of functions from a module, you can
selectively import just those functions.

If we wanted to import only the nub and sort functions from
Data.List, we’d do this:

import Data.List (nub, sort)

You can also choose to import all of the functions of a module
except a few select ones. That’s often useful when several modules
export functions with the same name and you want to get rid of
the offending ones. Say we already have our own function that’s
called nub and we want to import all the functions from
Data.List except the nub function:

import Data.List hiding (nub)



Importing modules

Another way of dealing with name clashes is to do qualified
imports.

The Data.Map module, which offers a data structure for looking
up values by key, exports a bunch of functions with the same name
as Prelude functions, like filter or null.

So when we import Data.Map and then call filter, Haskell won’t
know which function to use.

Here’s how we solve this:

import qualified Data.Map



Importing modules

import qualified Data.Map

This makes it so that if we want to reference Data.Map’s filter

function, we have to do Data.Map.filter, whereas just filter
still refers to the normal filter we all know (and love ,).

But typing out Data.Map in front of every function from that
module is kind of tedious. That’s why we can rename the qualified
import to something shorter:

import qualified Data.Map as M

Now, to reference Data.Map’s filter function, we just use
M.filter.



Modules

Use https://downloads.haskell.org/~ghc/latest/docs/

html/libraries/ to see which modules are in the standard
library.

To search for functions or to find out where they’re located, use
Hoogle (https://www.haskell.org/hoogle/). It’s a really
awesome Haskell search engine, you can search by name, module
name or even type signature.

https://downloads.haskell.org/~ghc/latest/docs/html/libraries/
https://downloads.haskell.org/~ghc/latest/docs/html/libraries/
https://www.haskell.org/hoogle/

Data.List

intersperse takes an element and a list and then puts that
element in between each pair of elements in the list.

Prelude Data.List> :t intersperse

intersperse :: a -> [a] -> [a]

Prelude Data.List> intersperse '.' "MONKEY"

"M.O.N.K.E.Y"

Prelude Data.List> intersperse 0 [1..10]

[1,0,2,0,3,0,4,0,5,0,6,0,7,0,8,0,9,0,10]

Prelude Data.List>



Data.List

intercalate takes a list of lists and a list. It then inserts that list
in between all those lists and then flattens the result.

Prelude Data.List> :t intercalate

intercalate :: [a] -> [[a]] -> [a]

Prelude Data.List> intercalate " " ["hey","there","guys"]

"hey there guys"

Prelude Data.List> intercalate [0,0] [[1,2],[3,4,5],[6,7]]

[1,2,0,0,3,4,5,0,0,6,7]

Prelude Data.List>



Data.List

transpose transposes a list of lists. If you look at a list of lists as
a 2D matrix, the columns become the rows and vice versa.

Prelude Data.List> :t transpose

transpose :: [[a]] -> [[a]]

Prelude Data.List> transpose [[1,2,3],[4,5,6],[7,8,9]]

[[1,4,7],[2,5,8],[3,6,9]]

Prelude Data.List> transpose ["hey","there","guys"]

["htg","ehu","yey","rs","e"]

Prelude Data.List>



Data.List

concat flattens a list of lists into just a list of elements.

Prelude Data.List> :t concat

concat :: [[a]] -> [a]

Prelude Data.List> concat ["foo","bar","car"]

"foobarcar"

Prelude Data.List> concat [[3,4,5],[2,3,4],[2,1,1]]

[3,4,5,2,3,4,2,1,1]

Prelude Data.List>

It will just remove one level of nesting. So if you want to
completely flatten [[[2,3],[3,4,5],[2]],[[2,3],[3,4]]],
which is a list of lists of lists, you have to concatenate it twice.



Data.List

Doing concatMap is the same as first mapping a function to a list
and then concatenating the list with concat.

Prelude Data.List> :t concatMap

concatMap :: (a -> [b]) -> [a] -> [b]

Prelude Data.List> concatMap (replicate 4) [1..3]

[1,1,1,1,2,2,2,2,3,3,3,3]

Prelude Data.List> concatMap (\x -> [x]) [1..3]

[1,2,3]

Prelude Data.List> concatMap (\x -> [[x]]) [1..3]

[[1],[2],[3]]

Prelude Data.List>



Data.List

and takes a list of boolean values and returns True only if all the
values in the list are True.

Prelude Data.List> :t and

and :: [Bool] -> Bool

Prelude Data.List> and $ map (>4) [5,6,7,8]

True

Prelude Data.List> and $ map (==4) [4,4,4,3,4]

False

Prelude Data.List> and $ map ('a' `elem`) ["to", "ti", "ta"]

False

Prelude Data.List> and $ map ('t' `elem`) ["to", "ti", "ta"]

True

Prelude Data.List>



Data.List

or is like and, only it returns True if any of the boolean values in
a list is True.

Prelude Data.List> :t or

or :: [Bool] -> Bool

Prelude Data.List> or $ map (==4) [2,3,4,5,6,1]

True

Prelude Data.List> or $ map (>4) [1,2,3]

False

Prelude Data.List>



Data.List

any and all take a predicate and then check if any or all the
elements in a list satisfy the predicate, respectively. Usually we use
these two functions instead of mapping over a list and then doing
and or or.

Prelude Data.List> :t any

any :: (a -> Bool) -> [a] -> Bool

Prelude Data.List> any (==4) [2,3,5,6,1,4]

True

Prelude Data.List> any (`elem` ['A'..'Z']) "HEYGUYSwhatsup"

True

Prelude Data.List> :t all

all :: (a -> Bool) -> [a] -> Bool

Prelude Data.List> all (>4) [6,9,10]

True

Prelude Data.List> all (`elem` ['A'..'Z']) "HEYGUYSwhatsup"

False

Prelude Data.List>



Data.List

iterate takes a function and a starting value. It applies the
function to the starting value, then it applies that function to the
result, then it applies the function to that result again, etc. It
returns all the results in the form of an infinite list.

Prelude Data.List> :t iterate

iterate :: (a -> a) -> a -> [a]

Prelude Data.List> take 10 $ iterate (*2) 1

[1,2,4,8,16,32,64,128,256,512]

Prelude Data.List> take 3 $ iterate (++ "haha") "haha"

["haha","hahahaha","hahahahahaha"]

Prelude Data.List>



Data.List

splitAt takes a number and a list. It then splits the list at that
many elements, returning the resulting two lists in a tuple.

Prelude Data.List> :t splitAt

splitAt :: Int -> [a] -> ([a], [a])

Prelude Data.List> splitAt 3 "heyman"

("hey","man")

Prelude Data.List> splitAt 100 "heyman"

("heyman","")

Prelude Data.List> splitAt (-3) "heyman"

("","heyman")

Prelude Data.List> let (a,b) = splitAt 3 "foobar" in b ++ a

"barfoo"

Prelude Data.List>



Data.List

takeWhile is a really useful little function. It takes elements from
a list while the predicate holds and then when an element is
encountered that doesn’t satisfy the predicate, it’s cut off. It turns
out this is very useful.

Prelude Data.List> :t takeWhile

takeWhile :: (a -> Bool) -> [a] -> [a]

Prelude Data.List> takeWhile (>3) [6,5,4,3,2,1,2,3,4,5,4,3,2,1]

[6,5,4]

Prelude Data.List> takeWhile (/=' ') "This is a sentence"

"This"

Prelude Data.List> sum $ takeWhile (<10000) $ map (^3) [1..]

53361

Prelude Data.List>



Data.List

dropWhile is similar to takeWhile, only it drops all the elements
while the predicate is true. Once predicate equates to False, it
returns the rest of the list. An extremely useful (and lovely ,)
function!

Prelude Data.List> :t dropWhile

dropWhile :: (a -> Bool) -> [a] -> [a]

Prelude Data.List> dropWhile (/=' ') "This is a sentence"

" is a sentence"

Prelude Data.List> dropWhile (<3) [1,2,2,2,3,4,5,4,3,2,1]

[3,4,5,4,3,2,1]

Prelude Data.List> :{

Prelude Data.List| let stock = [(9.2,2008,01),(11.4,2008,02),

Prelude Data.List| (7.2,2007,03)]

Prelude Data.List| :}

Prelude Data.List> head $ dropWhile (\(v,y,n) -> v < 10) stock

(11.4,2008,2)

Prelude Data.List>



Data.List

sort simply sorts a list. The type of the elements in the list has to
be part of the Ord typeclass, because if the elements of a list can’t
be put in some kind of order, then the list can’t be sorted.

Prelude Data.List> :t sort

sort :: Ord a => [a] -> [a]

Prelude Data.List> sort [8,5,3,2,1,6,4,2]

[1,2,2,3,4,5,6,8]

Prelude Data.List> sort ['a','g','d','b','d','c','f','e',]

<interactive>:204:39: parse error on input '']''

Prelude Data.List> sort ['a','g','d','b','d','c','f','e']

"abcddefg"

Prelude Data.List>

Data.Map

Association lists (also called dictionaries) are lists that are used to
store key-value pairs where ordering doesn’t matter.

For instance, we might use an association list to store phone
numbers, where phone numbers would be the values and people’s
names would be the keys. We don’t care in which order they’re
stored, we just want to get the right phone number for the right
person.

phoneBook =

[("betty","555-2938")

,("bonnie","452-2928")

,("patsy","493-2928")

,("lucille","205-2928")

,("wendy","939-8282")

,("penny","853-2492")

]



Data.Map

Let’s make a function that looks up some value given a key.

findKey :: (Eq k) => k -> [(k,v)] -> v

findKey k = snd . head . filter (\(k',v) -> k'==k)

Here, if a key isn’t in the association list, we’ll end up trying to get
the head of an empty list, which throws a runtime error. However,
we should avoid making our programs so easy to crash, so let’s use
the Maybe data type. If we don’t find the key, we’ll return a
Nothing. If we find it, we’ll return Just something, where
something is the value corresponding to that key.

findKey :: (Eq k) => k -> [(k,v)] -> Maybe v

findKey key [] = Nothing

findKey k ((k',v):xs) = if k == k'

then Just v

else findKey k xs



Data.Map

findKey :: (Eq k) => k -> [(k,v)] -> Maybe v

findKey k = foldr (\(k',v) acc -> if k==k' then Just v else acc) Nothing

It’s usually better to use folds for this standard list recursion
pattern instead of explicitly writing the recursion because they’re
easier to read and identify. Everyone knows it’s a fold when they
see the foldr call, but it takes some more thinking to read explicit
recursion.



Data.Map

ghci> findKey "penny" phoneBook

Just "853-2492"

ghci> findKey "betty" phoneBook

Just "555-2938"

ghci> findKey "wilma" phoneBook

Nothing

We just implemented the lookup function from Data.List.

If we want to find the corresponding value to a key, we have to
traverse all the elements of the list until we find it.

The Data.Map module offers association lists that are much faster
(because they’re internally implemented with trees) and also it
provides a lot of utility functions.



Data.Map

Because Data.Map exports functions that clash with the Prelude

and Data.List ones, we’ll do a qualified import.

import qualified Data.Map as Map



Data.Map

The fromList function takes an association list (in the form of a
list) and returns a map with the same associations.

Prelude Map> :t Map.fromList

Map.fromList :: Ord k => [(k, a)] -> Map.Map k a

Prelude Map> Map.fromList [('a', 1),('b', 2),('c',3),('d',4)]

fromList [('a',1),('b',2),('c',3),('d',4)]

Prelude Map> Map.fromList [(i,i^2) | i <- [1..5]]

fromList [(1,1),(2,4),(3,9),(4,16),(5,25)]

Prelude Map> Map.fromList [(1,1),(1,1),(2,2)]

fromList [(1,1),(2,2)]

Prelude Map>



Data.Map

empty represents an empty map. It takes no arguments, it just
returns an empty map.

Prelude Map> :t Map.empty

Map.empty :: Map.Map k a

Prelude Map> Map.empty

fromList []

Prelude Map>



Data.Map
insert

insert takes a key, a value and a map and returns a new map
that’s just like the old one, only with the key and value inserted.

Prelude Map> :t Map.insert

Map.insert :: Ord k => k -> a -> Map.Map k a -> Map.Map k a

Prelude Map> Map.insert 'a' 1 Map.empty

fromList [('a',1)]

Prelude Map> Map.insert 'a' 1 $ Map.insert 'b' 2 Map.empty

fromList [('a',1),('b',2)]

Prelude Map> let m = Map.empty

Prelude Map> Map.insert 'a' 1 $ Map.insert 'b' 2 m

fromList [('a',1),('b',2)]

Prelude Map>



Data.Map
insert

We can implement our own fromList by using the empty map,
insert and a fold.

fromList' :: (Ord k) => [(k,v)] -> Map.Map k v

fromList' = foldr (\(k,v) acc -> Map.insert k v acc) Map.empty

It’s a pretty straightforward fold. We start of with an empty map
and we fold it up from the right, inserting the key value pairs into
the accumulator as we go along.



Data.Map
null

null checks if a map is empty.

Prelude Map> :t Map.null

Map.null :: Map.Map k a -> Bool

Prelude Map> Map.null Map.empty

True

Prelude Map> Map.null $ Map.insert "k" "v" Map.empty

False

Prelude Map>



Data.Map
size

size reports the size of a map.

Prelude Map> :t Map.size

Map.size :: Map.Map k a -> Int

Prelude Map> Map.size Map.empty

0

Prelude Map> Map.size $ Map.insert "k" "v" Map.empty

1

Prelude Map>



Data.Map
singleton

singleton takes a key and a value and creates a map that has
exactly one mapping.

Prelude Map> :t Map.singleton

Map.singleton :: k -> a -> Map.Map k a

Prelude Map> Map.singleton 3 9

fromList [(3,9)]

Prelude Map> Map.insert 5 9 $ Map.singleton 3 9

fromList [(3,9),(5,9)]

Prelude Map>



Data.Map
lookup

lookup works like the Data.List lookup, only it operates on
maps. It returns Just something if it finds something for the key
and Nothing if it doesn’t.

Prelude Map> :t Map.lookup

Map.lookup :: Ord k => k -> Map.Map k a -> Maybe a

Prelude Map> let m = Map.insert 5 9 $ Map.singleton 3 9

Prelude Map> Map.lookup 5 m

Just 9

Prelude Map> Map.lookup 6 m

Nothing

Prelude Map>



Data.Map
member

member is a predicate takes a key and a map and r eports whether
the key is in the map or not.

Prelude Map> :t Map.member

Map.member :: Ord k => k -> Map.Map k a -> Bool

Prelude Map> let m = Map.insert 5 9 $ Map.singleton 3 9

Prelude Map> Map.member 5 m

True

Prelude Map> Map.member 6 m

False

Prelude Map>



Data.Map
map and filter

map and filter work much like their list equivalents.

Prelude Map> let m = Map.fromList [(1,1),(2,4),(3,9)]

Prelude Map> :t Map.map

Map.map :: (a -> b) -> Map.Map k a -> Map.Map k b

Prelude Map> Map.map (*100) m

fromList [(1,100),(2,400),(3,900)]

Prelude Map> :t Map.filter

Map.filter :: (a -> Bool) -> Map.Map k a -> Map.Map k a

Prelude Map> Map.filter (>2) m

fromList [(2,4),(3,9)]

Prelude Map>



Data.Map
toList

fromList is the inverse of fromList.

Prelude Map> :t Map.toList

Map.toList :: Map.Map k a -> [(k, a)]

Prelude Map> Map.toList . Map.insert 9 2 $ Map.singleton 4 3

[(4,3),(9,2)]

Prelude Map> Map.toList Map.empty

[]

Prelude Map>



Data.Map
keys and elems

keys and elems return lists of keys and values respectively. keys

is the equivalent of map fst . Map.toList and elems is the
equivalent of map snd . Map.toList.

Prelude Map> :t Map.keys

Map.keys :: Map.Map k a -> [k]

Prelude Map> :t Map.elems

Map.elems :: Map.Map k a -> [a]

Prelude Map> Map.keys $ Map.insert 9 2 $ Map.singleton 4 3

[4,9]

Prelude Map> Map.elems $ Map.insert 9 2 $ Map.singleton 4 3

[3,2]

Prelude Map>



Data.Set

The Data.Set module offers us, well, sets.

Sets are kind of like a cross between lists and maps. All the
elements in a set are unique. And because they’re internally
implemented with trees (much like maps in Data.Map), they’re
ordered.

Checking for membership, inserting, deleting, etc. is much faster
than doing the same thing with lists.

The most common operation when dealing with sets are inserting
into a set, checking for membership and converting a set to a list.

Because the names in Data.Set clash with a lot of Prelude and
Data.List names, we do a qualified import (e.g.,
import qualified Data.Set as Set).



Data.Set
fromList

The fromList function works much like you would expect. It
takes a list and converts it into a set.

ghci> :t Set.fromList

Set.fromList :: Ord a => [a] -> Data.Set.Set a

ghci> Set.fromList []

fromList []

ghci> Set.fromList "i'm a poor lonesome cowboy"

fromList " 'abceilmnoprswy"

ghci>



Data.Set
intersection

Use intersection function to see which elements sets both share.

ghci> let set1 = Set.fromList "To be or not to Be"

ghci> let set2 = Set.fromList "That is the question"

ghci> :t Set.intersection

Set.intersection

:: Ord a => Data.Set.Set a -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.intersection set1 set2

fromList " Tenot"

ghci>



Data.Set
difference

We can use the difference function to see which letters are in
the first set but aren’t in the second one and vice versa.

ghci> let set1 = Set.fromList "To be or not to Be"

ghci> let set2 = Set.fromList "That is the question"

ghci> :t Set.difference

Set.difference

:: Ord a => Data.Set.Set a -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.difference set1 set2

fromList "Bbr"

ghci> Set.difference set2 set1

fromList "ahiqsu"

ghci>



Data.Set
union

we can see all the unique letters used in both sentences by using
union.

ghci> let set1 = Set.fromList "To be or not to Be"

ghci> let set2 = Set.fromList "That is the question"

ghci> :t Set.union

Set.union

:: Ord a => Data.Set.Set a -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.union set1 set2

fromList " BTabehinoqrstu"

ghci>



Data.Set
null, size, . . .

The null, size, member, empty, singleton, insert and
delete functions all work like you’d expect them to.

ghci> :t Set.null

Set.null :: Data.Set.Set a -> Bool

ghci> :t Set.empty

Set.empty :: Data.Set.Set a

ghci> Set.null Set.empty

True

ghci>



Data.Set
null, size, . . .

The null, size, member, empty, singleton, insert and
delete functions all work like you’d expect them to.

ghci> Set.null $ Set.fromList [3,4,5,5,4,3]

False

ghci> :t Set.size

Set.size :: Data.Set.Set a -> Int

ghci> Set.size $ Set.fromList [3,4,5,3,4,5]

3

ghci>



Data.Set
null, size, . . .

The null, size, member, empty, singleton, insert and
delete functions all work like you’d expect them to.

ghci> :t Set.singleton

Set.singleton :: a -> Data.Set.Set a

ghci> Set.singleton 9

fromList [9]

ghci> Set.null $ Set.singleton 9

False

ghci> Set.size $ Set.singleton 9

1

ghci>



Data.Set
null, size, . . .

The null, size, member, empty, singleton, insert and
delete functions all work like you’d expect them to.

ghci> :t Set.insert

Set.insert :: Ord a => a -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.insert 4 $ Set.fromList [9,3,8,1]

fromList [1,3,4,8,9]

ghci> Set.insert 8 $ Set.fromList [5..10]

fromList [5,6,7,8,9,10]

ghci> Set.insert 8 $ Set.fromList [5..10]



Data.Set
null, size, . . .

The null, size, member, empty, singleton, insert and
delete functions all work like you’d expect them to.

ghci> :t Set.delete

Set.delete :: Ord a => a -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.delete 4 $ Set.fromList [3,4,5,4,3,4,5]

fromList [3,5]

ghci> Set.delete 1 $ Set.fromList [3,4,5,4,3,4,5]

fromList [3,4,5]

ghci>



Data.Set
isSubsetOf and isProperSubsetOf

We can also check for subsets or proper subset. Set A is a subset
of set B if B contains all the elements that A does. Set A is a
proper subset of set B if B contains all the elements that A does
but has more elements.

ghci> :t Set.isSubsetOf

Set.isSubsetOf :: Ord a => Data.Set.Set a -> Data.Set.Set a -> Bool

ghci> :t Set.isProperSubsetOf

Set.isProperSubsetOf

:: Ord a => Data.Set.Set a -> Data.Set.Set a -> Bool

ghci> Set.fromList [2,3,4] `Set.isSubsetOf` Set.fromList [1,2,3,4,5]

True

ghci> Set.fromList [1,2,3,4,5] `Set.isSubsetOf` Set.fromList [1,2,3,4,5]

True

ghci> Set.fromList [1,2,3,4,5] `Set.isProperSubsetOf` Set.fromList [1,2,3,4,5]

False

ghci> Set.fromList [2,3,4,8] `Set.isSubsetOf` Set.fromList [1,2,3,4,5]

False

ghci>



Data.Set
map and filter

We can also map over sets and filter them.

ghci> :t Set.map

Set.map :: Ord b => (a -> b) -> Data.Set.Set a -> Data.Set.Set b

ghci> Set.map (+1) $ Set.fromList [3,4,5,6,7,2,3,4]

fromList [3,4,5,6,7,8]

ghci> :t Set.filter

Set.filter :: (a -> Bool) -> Data.Set.Set a -> Data.Set.Set a

ghci> Set.filter odd $ Set.fromList [3,4,5,6,7,2,3,4]

fromList [3,5,7]

ghci>



Data.Set
Sets are often used to weed a list of duplicates from a list by first
making it into a set with fromList and then converting it back to
a list with toList.

The Data.List function nub already does that, but weeding out
duplicates for large lists is much faster if you cram them into a set
and then convert them back to a list than using nub.

But using nub only requires the type of the list’s elements to be
part of the Eq typeclass, whereas if you want to cram elements
into a set, the type of the list has to be in Ord.

ghci> let setNub xs = Set.toList $ Set.fromList xs

ghci> setNub "HEY WHATS CRACKALACKIN"

" ACEHIKLNRSTWY"

ghci> import Data.List as List

ghci> List.nub "HEY WHATS CRACKALACKIN"

"HEY WATSCRKLIN"

ghci>



Making you own module



Making you own module

At the beginning of a module, we specify the module name.

If we have a file called Geometry.hs, then we should name our
module Geometry. Then, we specify the functions that it exports
and after t hat, we can start writing the functions.

module Geometry

( sphereVolume

, sphereArea

, cubeVolume

, cubeArea

, cuboidArea

, cuboidVolume

) where

...



Making you own module

sphereVolume :: Float -> Float

sphereVolume radius = (4.0 / 3.0) * pi * (radius ^ 3)

sphereArea :: Float -> Float

sphereArea radius = 4 * pi * (radius ^ 2)

cubeVolume :: Float -> Float

cubeVolume side = cuboidVolume side side side

cubeArea :: Float -> Float

cubeArea side = cuboidArea side side side

cuboidVolume :: Float -> Float -> Float -> Float

cuboidVolume a b c = rectangleArea a b * c

cuboidArea :: Float -> Float -> Float -> Float

cuboidArea a b c = rectangleArea a b * 2 + rectangleArea a c * 2 +

rectangleArea c b * 2

-- not exported

rectangleArea :: Float -> Float -> Float

rectangleArea a b = a * b



Making you own module

When making a module, we usually export only those functions
that act as a sort of interface to our module so that the
implementation is hidden.

If someone is using our Geometry module, they don’t have to
concern themselves with functions that we don’t export.

We can decide to change those functions completely or delete
them in a newer version (we could delete rectangleArea and just
use * instead) and no one will mind because we weren’t exporting
them in the first place.

To use our module:

import Geometry

or

import qualified Geometry

or

import qualified Geometry as G

Geometry.hs has to be in the same folder that the program that’s
importing it is in, though.



Making you own module

Modules can also be given a hierarchical structures.

Each module can have a number of sub-modules and they can
have sub-modules of their own.

Let’s section these functions off so that Geometry is a module that
has three sub-modules, one for each type of object.

First, we’ll make a folder called Geometry. Mind the capital G. In
it, we’ll place three files: Sphere.hs, Cuboid.hs, and Cube.hs.



Making you own module
Sphere.hs

module Geometry.Sphere

( volume

, area

) where

volume :: Float -> Float

volume radius = (4.0 / 3.0) * pi * (radius ^ 3)

area :: Float -> Float

area radius = 4 * pi * (radius ^ 2)



Making you own module
Cuboid.hs

module Geometry.Cuboid

( volume

, area

) where

volume :: Float -> Float -> Float -> Float

volume a b c = rectangleArea a b * c

area :: Float -> Float -> Float -> Float

area a b c = rectangleArea a b * 2 + rectangleArea a c * 2 +

rectangleArea c b * 2

rectangleArea :: Float -> Float -> Float

rectangleArea a b = a * b



Making you own module
Cube.hs

module Geometry.Cube

( volume

, area

) where

import qualified Geometry.Cuboid as Cuboid

volume :: Float -> Float

volume side = Cuboid.volume side side side

area :: Float -> Float

area side = Cuboid.area side side side



Making you own module

So now if we’re in a file that’s on the same level as the Geometry

folder, we can do, say:

import Geometry.Sphere

If we want to juggle two or more of these modules, we have to do
qualified imports because they export functions with the same
names. So we just do something like:

import qualified Geometry.Sphere as Sphere

import qualified Geometry.Cuboid as Cuboid

import qualified Geometry.Cube as Cube


